

# Chapter 1

## Calculating an Exceptional Machine

Graham Hutton and Joel Wright<sup>1</sup>

**Abstract:** In previous work we showed how to verify a compiler for a small language with exceptions. In this article we show how to *calculate*, as opposed to verify, an abstract machine for this language. The key step is the use of Reynold’s *defunctionalization*, an old program transformation technique that has recently been rejuvenated by the work of Danvy et al.

### 1.1 INTRODUCTION

Exceptions are an important feature of modern programming languages, but their compilation has traditionally been viewed as an advanced topic. In previous work we showed how the basic method of compiling exceptions using *stack unwinding* can be explained and verified using elementary functional programming techniques [HW04]. In particular, we developed a compiler for a small language with exceptions, together with a proof of its correctness.

In the formal reasoning community, however, one prefers *constructions* to verifications [Bac03]. That is, rather than first writing the compiler and then separately proving its correctness with respect to a semantics for the language, it would be preferable to try and calculate the compiler [Mei92] directly from the semantics, with the aim of giving a systematic *discovery* of the idea of compiling exceptions using stack unwinding, as opposed to a post-hoc verification.

In this article we take a step towards this goal, by showing how to calculate an abstract machine for evaluating expressions in our language with exceptions. The key step in the calculation is the use of *defunctionalization*, a program transformation technique that eliminates the use of higher-order functions, first introduced by Reynolds in his seminal work on definitional interpreters [Rey72].

---

<sup>1</sup>School of Computer Science and IT, University of Nottingham, Jubilee Campus, Wollaton Road, Nottingham NG8 1BB, UK. Email: {gmh,jjw}@cs.nott.ac.uk.

Despite being simple and powerful, defunctionalization seems to be somewhat neglected in recent years. For example, it features in few modern courses, textbooks, and research articles on program transformation, and does not seem to be as widely known and used as it should be. Recently, however, defunctionalization has been rejuvenated by the work of Danvy et al, who show how it can be applied in a variety of different areas, including the systematic design of abstract machines for functional languages [DN01, ABDM03b, ADM04].

In this article, we show how Danvy’s approach can be used to calculate an abstract machine for our language with exceptions. Moreover, the calculation is *rabbit free*, in the sense that there are no Eureka steps in which one needs to metaphorically pull a rabbit out of a hat — all the required concepts arise naturally from the calculation process itself. The approach is based upon the work of Danvy et al, but the emphasis on calculation and the style of exposition are our own.

The language that we use comprises just integer values, an addition operator, a single exceptional value called `throw`, and a `catch` operator for this value [HW04]. This language does not provide features that are necessary for actual programming, but it *does* provide just what we need for expository purposes. In particular, integers and addition constitute a minimal language in which to consider computation using a stack, and `throw` and `catch` constitute a minimal extension in which such computations can involve exceptions.

Our development proceeds in two steps, starting with the exception-free part of the language to introduce the basic techniques, to which support for exceptions is then added in the second step. All the programs are written in Haskell [Pey03], and all the calculations are presented using equational reasoning. An extended version of the article that includes the calculations omitted here for reasons of space is available from [www.cs.nott.ac.uk/~gmh/machine-extended.pdf](http://www.cs.nott.ac.uk/~gmh/machine-extended.pdf).

## 1.2 ABSTRACT MACHINES

An *abstract machine* can be defined as a term rewriting system for executing programs in a particular language, and is given by a set of rewrite rules that make explicit how each step of execution proceeds. Perhaps the best known example is Landin’s SECD machine for the lambda calculus [Lan64], which comprises a set of rewrite rules that operate on tuples with four components that give the machine its name, called the stack, environment, control and dump.

For a simpler example, consider a language in which programs comprise a sequence of push and add operations on a stack of integers. In Haskell, such programs, operations and stacks can be represented by the following types:

```
type Prog  = [Op]
data Op    = PUSH Int | ADD
type Stack = [Int]
```

An abstract machine for this language is given by defining two rewrite rules on pairs of programs and stacks from the set  $Prog \times Stack$ :

$$\begin{aligned} \langle PUSH\ n : ops , s \rangle &\longrightarrow \langle ops , n : s \rangle \\ \langle ADD : ops , n : m : s \rangle &\longrightarrow \langle ops , n + m : s \rangle \end{aligned}$$

The first rule states that push places a new integer on top of the stack, while the second states that add replaces the top two integers on the stack by their sum. This machine can be implemented in Haskell by an execution function that repeatedly applies the two rules until this is no longer possible:

$$\begin{aligned} exec &:: (Prog, Stack) \rightarrow (Prog, Stack) \\ exec (PUSH\ n : ops, s) &= exec (ops, n : s) \\ exec (ADD : ops, n : m : s) &= exec (ops, n + m : s) \\ exec (p, s) &= (p, s) \end{aligned}$$

For example,  $exec ([PUSH\ 1, PUSH\ 2, ADD], [])$  gives the result  $([], [3])$ . In the remainder of this article, we will use the term abstract machine for such a functional implementation of an underlying set of rewrite rules.

### 1.3 ARITHMETIC EXPRESSIONS

As in our previous work [HW04], let us begin our development by considering a simple language of expressions comprising integers and addition, whose semantics is given by a function that evaluates an expression to its integer value:

$$\begin{aligned} \mathbf{data}\ Expr &= Val\ Int \mid Add\ Expr\ Expr \\ eval &:: Expr \rightarrow Int \\ eval (Val\ n) &= n \\ eval (Add\ x\ y) &= eval\ x + eval\ y \end{aligned}$$

We will now calculate an abstract machine for this language, by making a series of three transformations to the semantics.

#### Step 1 - Add continuations

At present, the order in which addition evaluates its argument expressions is determined by the language in which the semantics is written, in this case Haskell. The first step in producing an abstract machine is to make the order of evaluation explicit in the semantics itself. A standard technique for achieving this aim is to rewrite the semantics in *continuation-passing* style [Rey72].

A *continuation* is a function that will be applied to the result of an evaluation. For example, in the equation  $eval (Add\ x\ y) = eval\ x + eval\ y$  from our semantics, when the first recursive call,  $eval\ x$ , is being evaluated, the remainder of the right-hand side,  $+ eval\ y$ , can be viewed as a continuation for this evaluation, in the sense that it is the function that will be applied to the result.

More formally, in the context of our semantics  $eval :: Expr \rightarrow Int$ , a continuation is a function of type  $Int \rightarrow Int$  that will be applied to the result of type  $Int$  to give a new result of type  $Int$ . (This type can be generalised to  $Int \rightarrow a$ , but we

don't need the extra generality for our purposes here.) We capture the notion of such a continuation using the following type definition:

$$\mathbf{type} \textit{Cont} = \textit{Int} \rightarrow \textit{Int}$$

Our aim now is to define a new semantics,  $eval'$ , that takes an expression and returns an integer as previously, but also takes a continuation that will be applied to the resulting integer. That is, we seek to define a function

$$eval' \quad :: \quad \textit{Expr} \rightarrow \textit{Cont} \rightarrow \textit{Int}$$

such that:

$$eval' e c = c (eval e)$$

At this point in most texts, a recursive definition for  $eval'$  would normally be written and then either proved to satisfy the above equation, or this be justified by appealing to the correctness of a general continuation-passing transformation. However, we prefer to *calculate* the definition for  $eval'$  directly from the above equation, by the use of structural induction on  $\textit{Expr}$ .

Case:  $\textit{Val} n$

$$\begin{aligned} & eval' (\textit{Val} n) c \\ = & \quad \{ \textit{specification of } eval' \} \\ & c (eval (\textit{Val} n)) \\ = & \quad \{ \textit{definition of } eval \} \\ & c n \end{aligned}$$

Case:  $\textit{Add} x y$

$$\begin{aligned} & eval' (\textit{Add} x y) c \\ = & \quad \{ \textit{specification of } eval' \} \\ & c (eval (\textit{Add} x y)) \\ = & \quad \{ \textit{definition of } eval \} \\ & c (eval x + eval y) \\ = & \quad \{ \textit{abstraction over } eval x \} \\ & (\lambda n \rightarrow c (n + eval y)) (eval x) \\ = & \quad \{ \textit{induction hypothesis for } x \} \\ & eval' x (\lambda n \rightarrow c (n + eval y)) \\ = & \quad \{ \textit{abstraction over } eval y \} \\ & eval' x (\lambda n \rightarrow (\lambda m \rightarrow c (n + m))) (eval y) \\ = & \quad \{ \textit{induction hypothesis for } y \} \\ & eval' x (\lambda n \rightarrow eval' y (\lambda m \rightarrow c (n + m))) \end{aligned}$$

In conclusion, we have calculated the following recursive definition:

$$\begin{aligned} eval' & \quad :: \quad \textit{Expr} \rightarrow \textit{Cont} \rightarrow \textit{Int} \\ eval' (\textit{Val} n) c & \quad = \quad c n \\ eval' (\textit{Add} x y) c & \quad = \quad eval' x (\lambda n \rightarrow eval' y (\lambda m \rightarrow c (n + m))) \end{aligned}$$

That is, for an integer value we simply apply the continuation, while for an addition we evaluate the first argument and call the result  $n$ , then evaluate the second argument and call the result  $m$ , and finally apply the continuation to the sum of  $n$  and  $m$ . In this manner, order of evaluation is now explicit in the semantics.

Note that we have ensured that addition evaluates its arguments from left-to-right by first abstracting over  $eval\ x$  in the above calculation, and then abstracting over  $eval\ y$ . It would be perfectly valid to proceed in the other direction, which would result in right-to-left evaluation. Note also that our original semantics can be recovered from our new semantics, by substituting the identity continuation  $\lambda n \rightarrow n$  into the equation from which  $eval'$  was constructed. That is, our original semantics  $eval$  can now be redefined as follows:

$$\begin{aligned} eval &:: Expr \rightarrow Int \\ eval\ e &= eval'\ e\ (\lambda n \rightarrow n) \end{aligned}$$

## Step 2 - Defunctionalize

We have now taken a step towards an abstract machine by making evaluation order explicit, but in so doing have also taken a step away from such a machine by making the semantics into a higher-order function. The next step is to regain the first-order nature of the original semantics by eliminating the use of continuations, but retaining the explicit order of evaluation that they introduced.

A standard technique for eliminating the use of functions as arguments is *defunctionalization* [Rey72]. This technique is based upon the observation that we don't usually need the entire function-space of possible argument functions, because only a few forms of such functions are actually used in practice. Hence, we can represent the argument functions that we actually need using a datatype, rather than using the actual functions themselves.

In our new semantics, there are only three forms of continuations that are actually used, namely one to invoke the semantics, and two in the case for evaluating an addition. We begin by separating out these three forms, by giving them names and abstracting over their free variables. That is, we define three combinators for constructing the required forms of continuations:

$$\begin{aligned} c1 &:: Cont \\ c1 &= \lambda n \rightarrow n \\ c2 &:: Expr \rightarrow Cont \rightarrow Cont \\ c2\ y\ c &= \lambda n \rightarrow eval'\ y\ (c3\ n\ c) \\ c3 &:: Int \rightarrow Cont \rightarrow Cont \\ c3\ n\ c &= \lambda m \rightarrow c\ (n + m) \end{aligned}$$

At present we have just used anonymous names  $c1$ ,  $c2$  and  $c3$  for the combinators, but these will be replaced by more suggestive names later on. Using these

combinators, our semantics can now be rewritten as follows:

$$\begin{aligned}
eval' & :: Expr \rightarrow Cont \rightarrow Int \\
eval' (Val n) c & = c n \\
eval' (Add x y) c & = eval' x (c2 y c) \\
eval & :: Expr \rightarrow Int \\
eval e & = eval' e c1
\end{aligned}$$

The next stage in applying defunctionalization is to define a datatype whose values represent the three combinators that we have isolated:

$$\mathbf{data} \text{ CONT} = C1 | C2 Expr CONT | C3 Int CONT$$

The constructors of this datatype have the same types as the corresponding combinators, except that the new type *CONT* plays the role of *Cont*:

$$\begin{aligned}
C1 & :: CONT \\
C2 & :: Expr \rightarrow CONT \rightarrow CONT \\
C3 & :: Int \rightarrow CONT \rightarrow CONT
\end{aligned}$$

The fact that values of type *CONT* represent continuations of type *Cont* can be formalised by defining a function that maps from one to the other:

$$\begin{aligned}
apply & :: CONT \rightarrow Cont \\
apply C1 & = c1 \\
apply (C2 y c) & = c2 y (apply c) \\
apply (C3 n c) & = c3 n (apply c)
\end{aligned}$$

The name of this function derives from the fact that when its type is expanded to  $apply :: CONT \rightarrow Int \rightarrow Int$ , it can be viewed as applying a representation of a continuation to an integer to give another integer.

Our aim now is to define a new semantics,  $eval''$ , that behaves in the same way as our previous semantics,  $eval'$ , except that it uses values of type *CONT* rather than continuations of type *Cont*. That is, we seek to define a function

$$eval'' :: Expr \rightarrow CONT \rightarrow Int$$

such that:

$$eval'' e c = eval' e (apply c)$$

As previously, we calculate the definition for the function  $eval''$  directly from this equation by the use of structural induction on *Expr*.

Case: *Val n*

$$\begin{aligned}
& eval'' (Val n) c \\
= & \{ \text{specification of } eval'' \} \\
& eval' (Val n) (apply c) \\
= & \{ \text{definition of } eval' \} \\
& apply c n
\end{aligned}$$

Case:  $Add\ x\ y$

$$\begin{aligned}
& eval'' (Add\ x\ y)\ c \\
= & \quad \{ \text{specification of } eval'' \} \\
& eval' (Add\ x\ y)\ (apply\ c) \\
= & \quad \{ \text{definition of } eval' \} \\
& eval' x\ (c2\ y\ (apply\ c)) \\
= & \quad \{ \text{definition of } apply \} \\
& eval' x\ (apply\ (C2\ y\ c)) \\
= & \quad \{ \text{induction hypothesis for } x \} \\
& eval'' x\ (C2\ y\ c)
\end{aligned}$$

In conclusion, we have calculated the following recursive definition:

$$\begin{aligned}
& eval'' & :: & Expr \rightarrow CONT \rightarrow Int \\
& eval'' (Val\ n)\ c & = & apply\ c\ n \\
& eval'' (Add\ x\ y)\ c & = & eval'' x\ (C2\ y\ c)
\end{aligned}$$

However, the definition for  $apply$  still refers to the previous semantics  $eval'$ , via its use of the combinator  $c2$ . We calculate a new definition for  $apply$  that refers to our new semantics instead by the use of case analysis on  $CONT$ .

Case:  $C1$

$$\begin{aligned}
& apply\ C1\ n \\
= & \quad \{ \text{definition of } apply \} \\
& c1\ n \\
= & \quad \{ \text{definition of } c1 \} \\
& n
\end{aligned}$$

Case:  $C2\ y\ c$

$$\begin{aligned}
& apply\ (C2\ y\ c)\ n \\
= & \quad \{ \text{definition of } apply \} \\
& c2\ y\ (apply\ c)\ n \\
= & \quad \{ \text{definition of } c2 \} \\
& eval' y\ (c3\ n\ (apply\ c)) \\
= & \quad \{ \text{definition of } apply \} \\
& eval' y\ (apply\ (C3\ n\ c)) \\
= & \quad \{ \text{specification of } eval'' \} \\
& eval'' y\ (C3\ n\ c)
\end{aligned}$$

Case:  $C3\ n\ c$

$$\begin{aligned}
& apply\ (C3\ n\ c)\ m \\
= & \quad \{ \text{definition of } apply \} \\
& c3\ n\ (apply\ c)\ m \\
= & \quad \{ \text{definition of } c3 \} \\
& apply\ c\ (n + m)
\end{aligned}$$

In conclusion, we have calculated the following new definition:

$$\begin{aligned} \mathit{apply} &:: \mathit{CONT} \rightarrow \mathit{Int} \rightarrow \mathit{Int} \\ \mathit{apply} \ C1 \ n &= n \\ \mathit{apply} \ (C2 \ y \ c) \ n &= \mathit{eval}'' \ y \ (C3 \ n \ c) \\ \mathit{apply} \ (C3 \ n \ c) \ m &= \mathit{apply} \ c \ (n + m) \end{aligned}$$

We have now eliminated the use of functions as arguments, and hence made the semantics first-order again. But what about the fact that  $\mathit{eval}''$  and  $\mathit{apply}$  are curried functions, and hence return functions as results? As is common practice, we do not view the use of functions as results as being higher-order, as it is not essential and can easily be eliminated if required by uncurrying.

Finally, our original semantics can be recovered from our new semantics by redefining  $\mathit{eval} \ e = \mathit{eval}'' \ e \ C1$ , as can be verified by a simple calculation:

$$\begin{aligned} &\mathit{eval} \ e \\ = & \{ \text{previous definition of } \mathit{eval} \} \\ &\mathit{eval}' \ e \ (\lambda n \rightarrow n) \\ = & \{ \text{definition of } \mathit{c1} \} \\ &\mathit{eval}' \ e \ c1 \\ = & \{ \text{definition of } \mathit{apply} \} \\ &\mathit{eval}' \ e \ (\mathit{apply} \ C1) \\ = & \{ \text{specification of } \mathit{eval}'' \} \\ &\mathit{eval}'' \ e \ C1 \end{aligned}$$

### Step 3 - Refactor

At this point, after making two transformations to the original semantics, the reader may be wondering what we have actually produced? In fact, we now have an abstract machine for evaluating expressions, but this only becomes clear after we *refactor* the definitions, in this simple case by just renaming the components. In detail, we rename  $\mathit{CONT}$  as  $\mathit{Cont}$ ,  $\mathit{C1}$  as  $\mathit{STOP}$ ,  $\mathit{C2}$  as  $\mathit{EVAL}$ ,  $\mathit{C3}$  as  $\mathit{ADD}$ ,  $\mathit{eval}''$  as  $\mathit{eval}$ ,  $\mathit{apply}$  as  $\mathit{exec}$ , and  $\mathit{eval}$  as  $\mathit{run}$  to give the following machine:

$$\begin{aligned} \mathbf{data} \ \mathit{Cont} &= \mathit{STOP} \mid \mathit{EVAL} \ \mathit{Expr} \ \mathit{Cont} \mid \mathit{ADD} \ \mathit{Int} \ \mathit{Cont} \\ \mathit{eval} &:: \mathit{Expr} \rightarrow \mathit{Cont} \rightarrow \mathit{Int} \\ \mathit{eval} \ (\mathit{Val} \ n) \ c &= \mathit{exec} \ c \ n \\ \mathit{eval} \ (\mathit{Add} \ x \ y) \ c &= \mathit{eval} \ x \ (\mathit{EVAL} \ y \ c) \\ \mathit{exec} &:: \mathit{Cont} \rightarrow \mathit{Int} \rightarrow \mathit{Int} \\ \mathit{exec} \ \mathit{STOP} \ n &= n \\ \mathit{exec} \ (\mathit{EVAL} \ y \ c) \ n &= \mathit{eval} \ y \ (\mathit{ADD} \ n \ c) \\ \mathit{exec} \ (\mathit{ADD} \ n \ c) \ m &= \mathit{exec} \ c \ (n + m) \\ \mathit{run} &:: \mathit{Expr} \rightarrow \mathit{Int} \\ \mathit{run} \ e &= \mathit{eval} \ e \ \mathit{STOP} \end{aligned}$$

We now explain the four parts of the abstract machine in turn:



- *Cont* is the type of *control stacks* for the machine, containing instructions that determine the behaviour of the machine after evaluating the current expression. The meaning of the three forms of instructions, *STOP*, *EVAL* and *ADD* will be explained shortly. Note that the type of control stacks could itself be refactored as an explicit list of instructions, as follows:

```

type Cont = [Inst]
data Inst = ADD Int | EVAL Expr

```

However, we prefer the original definition above because it only requires the definition of a single type rather than a pair of types.

- *eval* evaluates an expression in the context of a control stack. If the expression is an integer value, it is already fully evaluated, and we simply execute the control stack using this integer as an argument. If the expression is an addition, we evaluate the first argument, *x*, placing the instruction *EVAL y* on top of the current control stack to indicate that the second argument, *y*, should be evaluated once that of the first argument is completed.
- *exec* executes a control stack in the context of an integer argument. If the stack is empty, represented by the instruction *STOP*, we simply return the integer argument as the result of the execution. If the top of the stack is an instruction *EVAL y*, we evaluate the expression *y*, placing the instruction *ADD n* on top of the remaining stack to indicate that the current integer argument, *n*, should be added together with the result of evaluating *y* once this is completed. Finally, if the top of the stack is an instruction *ADD m*, evaluation of the two arguments of an addition is now complete, and we execute the remaining control stack in the context of the sum of the two resulting integers.
- *run* evaluates an expression to give an integer, by invoking *eval* with the given expression and the empty control stack as arguments.

The fact that our machine uses two mutually recursive functions, *eval* and *exec*, reflects the fact that it has two states, depending upon whether it is being driven by the structure of the expression (*eval*) or the control stack (*exec*). To illustrate the machine, here is how it evaluates  $(2 + 3) + 4$ :

```

run (Add (Add (Val 2) (Val 3)) (Val 4))
= eval (Add (Add (Val 2) (Val 3)) (Val 4)) STOP
= eval (Add (Val 2) (Val 3)) (EVAL (Val 4) STOP)
= eval (Val 2) (EVAL (Val 3) (EVAL (Val 4) STOP))
= exec (EVAL (Val 3) (EVAL (Val 4) STOP)) 2
= eval (Val 3) (ADD 2 (EVAL (Val 4) STOP))
= exec (ADD 2 (EVAL (Val 4) STOP)) 3
= exec (EVAL (Val 4) STOP) 5
= eval (Val 4) (ADD 5 STOP)
= exec (ADD 5 STOP) 4
= exec STOP 9
= 9

```

Note how the function *eval* proceeds downwards to the leftmost integer in the expression, maintaining a trail of the pending right-hand expressions on the control stack. In turn, the function *exec* then proceeds upwards through the trail, transferring control back to *eval* and performing additions as appropriate.

Readers familiar with Huet's *zipper* data structure for navigating around expressions [Hue97] may find it useful to note that our type *Cont* is a zipper data structure for *Expr*, specialised to the purpose of evaluating expressions. Moreover, this specialised zipper arose naturally here by a process of systematic calculation, and did not require any prior knowledge of this structure.

#### 1.4 ADDING EXCEPTIONS

Now let us extend our language of arithmetic expressions with simple primitives for throwing and catching an exception:

$$\mathbf{data} \textit{Expr} = \dots \mid \textit{Throw} \mid \textit{Catch Expr Expr}$$

Informally, *Throw* abandons the current computation and throws an exception, while *Catch* *x* *y* behaves as the expression *x* unless it throws an exception, in which case the catch behaves as the *handler* expression *y*. To formalise the meaning of these new primitives, we first recall the *Maybe* type:

$$\mathbf{data} \textit{Maybe} \ a = \textit{Nothing} \mid \textit{Just} \ a$$

That is, a value of type *Maybe* *a* is either *Nothing*, which we think of as an exceptional value, or has the form *Just* *x* for some *x* of type *a*, which we think of as a normal value [Spi90]. Using this type, our original semantics for expressions can be rewritten to take account of exceptions as follows:

$$\begin{aligned} \textit{eval} & :: \textit{Expr} \rightarrow \textit{Maybe Int} \\ \textit{eval} (\textit{Val} \ n) & = \textit{Just} \ n \\ \textit{eval} (\textit{Add} \ x \ y) & = \mathbf{case} \ \textit{eval} \ x \ \mathbf{of} \\ & \quad \textit{Nothing} \rightarrow \textit{Nothing} \\ & \quad \textit{Just} \ n \rightarrow \mathbf{case} \ \textit{eval} \ y \ \mathbf{of} \\ & \quad \quad \textit{Nothing} \rightarrow \textit{Nothing} \\ & \quad \quad \textit{Just} \ m \rightarrow \textit{Just} \ (n + m) \\ \textit{eval} (\textit{Throw}) & = \textit{Nothing} \\ \textit{eval} (\textit{Catch} \ x \ y) & = \mathbf{case} \ \textit{eval} \ x \ \mathbf{of} \\ & \quad \textit{Nothing} \rightarrow \textit{eval} \ y \\ & \quad \textit{Just} \ n \rightarrow \textit{Just} \ n \end{aligned}$$

We will now calculate an abstract machine from this extended semantics, by following the same three-step process as previously. That is, we first add continuations, then defunctionalize, and finally refactor the definitions.

### Step 1 - Add continuations

Because our semantics now returns a result of type *Maybe Int*, the type of continuations that we use must be modified accordingly:

$$\mathbf{type\ Cont} = \mathit{Maybe\ Int} \rightarrow \mathit{Maybe\ Int}$$

Our aim now is to define a new semantics

$$\mathit{eval}' \quad :: \quad \mathit{Expr} \rightarrow \mathit{Cont} \rightarrow \mathit{Maybe\ Int}$$

such that:

$$\mathit{eval}'\ e\ c = c\ (\mathit{eval}\ e)$$

That is, the new semantics behaves in the same way as *eval*, except that it applies a continuation to the result. As previously, we can calculate a recursive definition for *eval'* directly from this equation by structural induction on *Expr*:

$$\begin{aligned} \mathit{eval}' & \quad :: \quad \mathit{Expr} \rightarrow \mathit{Cont} \rightarrow \mathit{Maybe\ Int} \\ \mathit{eval}'\ (\mathit{Val}\ n)\ c & = c\ (\mathit{Just}\ n) \\ \mathit{eval}'\ (\mathit{Throw})\ c & = c\ \mathit{Nothing} \\ \mathit{eval}'\ (\mathit{Add}\ x\ y)\ c & = \mathit{eval}'\ x\ (\lambda x' \rightarrow \mathbf{case}\ x' \mathbf{of} \\ & \quad \mathit{Nothing} \rightarrow c\ \mathit{Nothing} \\ & \quad \mathit{Just}\ n \rightarrow \mathit{eval}'\ y\ (\lambda y' \rightarrow \mathbf{case}\ y' \mathbf{of} \\ & \quad \quad \mathit{Nothing} \rightarrow c\ \mathit{Nothing} \\ & \quad \quad \mathit{Just}\ m \rightarrow c\ (\mathit{Just}\ (n + m)))) \\ \mathit{eval}'\ (\mathit{Catch}\ x\ y)\ c & = \mathit{eval}'\ x\ (\lambda x' \rightarrow \mathbf{case}\ x' \mathbf{of} \\ & \quad \mathit{Nothing} \rightarrow \mathit{eval}'\ y\ c \\ & \quad \mathit{Just}\ n \rightarrow c\ (\mathit{Just}\ n)) \end{aligned}$$

(The above and subsequent omitted calculations are included in the extended version of the article.) In turn, our original semantics can be recovered by invoking our new semantics with the identity continuation. That is, we have

$$\begin{aligned} \mathit{eval} & \quad :: \quad \mathit{Expr} \rightarrow \mathit{Maybe\ Int} \\ \mathit{eval}\ e & = \mathit{eval}'\ e\ (\lambda x \rightarrow x) \end{aligned}$$

### Step 2 - Defunctionalize

Our new semantics uses four forms of continuations, namely one to invoke the semantics, two in the case for addition, and one in the case for catch. We define

four combinators for constructing these continuations:

$$\begin{aligned}
c1 &:: \text{Cont} \\
c1 &= \lambda x \rightarrow x \\
c2 &:: \text{Expr} \rightarrow \text{Cont} \rightarrow \text{Cont} \\
c2 \ y \ c &= \lambda x' \rightarrow \mathbf{case} \ x' \ \mathbf{of} \\
&\quad \text{Nothing} \rightarrow c \ \text{Nothing} \\
&\quad \text{Just } n \rightarrow \text{eval}' \ y \ (c3 \ n \ c) \\
c3 &:: \text{Int} \rightarrow \text{Cont} \rightarrow \text{Cont} \\
c3 \ n \ c &= \lambda y' \rightarrow \mathbf{case} \ y' \ \mathbf{of} \\
&\quad \text{Nothing} \rightarrow c \ \text{Nothing} \\
&\quad \text{Just } m \rightarrow c \ (\text{Just} \ (n + m)) \\
c4 &:: \text{Expr} \rightarrow \text{Cont} \rightarrow \text{Cont} \\
c4 \ y \ c &= \lambda x' \rightarrow \mathbf{case} \ x' \ \mathbf{of} \\
&\quad \text{Nothing} \rightarrow \text{eval}' \ y \ c \\
&\quad \text{Just } n \rightarrow c \ (\text{Just} \ n)
\end{aligned}$$

Using these combinators, our semantics can now be rewritten as follows:

$$\begin{aligned}
\text{eval}' &:: \text{Expr} \rightarrow \text{Cont} \rightarrow \text{Maybe Int} \\
\text{eval}' \ (\text{Val } n) \ c &= c \ (\text{Just } n) \\
\text{eval}' \ (\text{Throw}) \ c &= c \ \text{Nothing} \\
\text{eval}' \ (\text{Add } x \ y) \ c &= \text{eval}' \ x \ (c2 \ y \ c) \\
\text{eval}' \ (\text{Catch } x \ y) \ c &= \text{eval}' \ x \ (c4 \ y \ c) \\
\text{eval} &:: \text{Expr} \rightarrow \text{Maybe Int} \\
\text{eval} \ e &= \text{eval}' \ e \ c1
\end{aligned}$$

We now define a datatype to represent the four combinators, together with an application function that formalises the representation:

$$\begin{aligned}
\mathbf{data} \ \text{CONT} &= \text{C1} \mid \text{C2 Expr CONT} \mid \text{C3 Int Cont} \mid \text{C4 Expr CONT} \\
\text{apply} &:: \text{CONT} \rightarrow \text{Cont} \\
\text{apply} \ \text{C1} &= c1 \\
\text{apply} \ (\text{C2 } y \ c) &= c2 \ y \ (\text{apply} \ c) \\
\text{apply} \ (\text{C3 } n \ c) &= c3 \ n \ (\text{apply} \ c) \\
\text{apply} \ (\text{C4 } y \ c) &= c4 \ y \ (\text{apply} \ c)
\end{aligned}$$

Our aim now is to define a new semantics

$$\text{eval}'' \ :: \ \text{Expr} \rightarrow \text{CONT} \rightarrow \text{Maybe Int}$$

such that:

$$\text{eval}'' \ e \ c = \text{eval}' \ e \ (\text{apply} \ c)$$

That is, the new semantics behaves in the same way as  $\text{eval}'$ , except that it uses representations of continuations rather than actual continuations. We can calculate

the definition for  $eval''$  by structural induction on  $Expr$ :

$$\begin{aligned}
eval'' & :: Expr \rightarrow CONT \rightarrow Maybe Int \\
eval'' (Val n) c & = apply c (Just n) \\
eval'' (Throw) c & = apply c Nothing \\
eval'' (Add x y) c & = eval'' x (C2 y c) \\
eval'' (Catch x y) c & = eval'' x (C4 y c)
\end{aligned}$$

In turn, we can calculate a new definition for  $apply$  by case analysis:

$$\begin{aligned}
apply & :: CONT \rightarrow Maybe Int \rightarrow Maybe Int \\
apply C1 x & = x \\
apply (C2 y c) Nothing & = apply c Nothing \\
apply (C2 y c) (Just n) & = eval'' y (C3 n c) \\
apply (C3 n c) Nothing & = apply c Nothing \\
apply (C3 n c) (Just m) & = apply c (Just (n + m)) \\
apply (C4 y c) Nothing & = eval'' y c \\
apply (C4 y c) (Just n) & = apply c (Just n)
\end{aligned}$$

Our original semantics can be recovered by invoking our new semantics with the representation of the identity continuation:

$$\begin{aligned}
eval & :: Expr \rightarrow Maybe Int \\
eval & = eval'' e C1
\end{aligned}$$

### Step 3 - Refactor

We now rename the components in the same way as previously, and rename the new combinator  $C4$  as  $HAND$ . This time around, however, refactoring amounts to more than just renaming. In particular, we split the application function

$$apply :: Cont \rightarrow Maybe Int \rightarrow Maybe Int$$

into two separate application functions

$$\begin{aligned}
exec & :: Cont \rightarrow Int \rightarrow Maybe Int \\
unwind & :: Cont \rightarrow Maybe Int
\end{aligned}$$

such that:

$$\begin{aligned}
apply c (Just n) & = exec c n \\
apply c Nothing & = unwind c
\end{aligned}$$

That is,  $exec$  deals with normal arguments, and  $unwind$  with exceptional arguments. We can calculate the definitions for  $exec$  and  $unwind$  by structural induc-

tion on *Cont*, as a result of which we obtain the following machine:

<b>data</b> <i>Cont</i>	=	<i>STOP</i>   <i>EVAL Expr Cont</i>   <i>ADD Int Cont</i>   <i>HAND Expr Cont</i>
<i>eval</i>	::	<i>Expr</i> → <i>Cont</i> → <i>Maybe Int</i>
<i>eval</i> ( <i>Val n</i> ) <i>c</i>	=	<i>exec c n</i>
<i>eval</i> ( <i>Throw</i> ) <i>c</i>	=	<i>unwind c</i>
<i>eval</i> ( <i>Add x y</i> ) <i>c</i>	=	<i>eval x (EVAL y c)</i>
<i>eval</i> ( <i>Catch x y</i> ) <i>c</i>	=	<i>eval x (HAND y c)</i>
<i>exec</i>	::	<i>Cont</i> → <i>Int</i> → <i>Maybe Int</i>
<i>exec STOP n</i>	=	<i>Just n</i>
<i>exec (EVAL y c) n</i>	=	<i>eval y (ADD n c)</i>
<i>exec (ADD n c) m</i>	=	<i>exec c (n + m)</i>
<i>exec (HAND _ c) n</i>	=	<i>exec c n</i>
<i>unwind</i>	::	<i>Cont</i> → <i>Maybe Int</i>
<i>unwind STOP</i>	=	<i>Nothing</i>
<i>unwind (EVAL _ c)</i>	=	<i>unwind c</i>
<i>unwind (ADD _ c)</i>	=	<i>unwind c</i>
<i>unwind (HAND y c)</i>	=	<i>eval y c</i>
<i>run</i>	::	<i>Expr</i> → <i>Maybe Int</i>
<i>run e</i>	=	<i>eval e STOP</i>

We now explain the three main functions of the abstract machine:

- *eval* evaluates an expression in the context of a control stack. The cases for integer values and addition are as previously. If the expression is a throw, we *unwind the stack* seeking a handler expression. If the expression is a catch, we evaluate its first argument, *x*, and *mark the stack* with the instruction *HAND y* to indicate that its second argument, the handler *y*, should be used if evaluation of its first produces an exceptional value.
- *exec* executes a control stack in the context of an integer argument. The first three cases are as previously, except that if the stack is empty the resulting integer is tagged as a normal result value. If the top of the stack is a handler instruction, there is no need for the associated handler expression because a normal integer result has already been produced, and we *unmark the stack* by popping the handler and then continue executing.
- *unwind* executes the control stack in the context of an exception. If the stack is empty, the exception is uncaught and we simply return the exceptional result value. If the top of the stack is an evaluation or an addition instruction, there is no need for their arguments because a handler is being sought, and we pop them from the stack and then continue unwinding. If the top of the stack is a handler instruction, we catch the exception by evaluating the associated handler expression in the context of the remaining stack.

Note that the idea of marking, unmarking, and unwinding the stack arose directly from the calculations, and did not require any prior knowledge of these concepts. It is also interesting to note that the above machine produced by calculation is both simpler and more efficient than those we had previously designed by hand. In particular, our previous machines did not make a clean separation between the three concepts of evaluating an expression (*eval*), executing the control stack (*exec*) and unwinding the control stack (*unwind*).

To illustrate our machine, here is how it evaluates  $1 + (\text{catch } (2 + \text{throw}) 3)$ :

$$\begin{aligned}
& \text{run } (\text{Add } (\text{Val } 1) (\text{Catch } (\text{Add } (\text{Val } 2) \text{Throw}) (\text{Val } 3))) \\
= & \text{eval } (\text{Add } (\text{Val } 1) (\text{Catch } (\text{Add } (\text{Val } 2) \text{Throw}) (\text{Val } 3))) \text{STOP} \\
= & \text{eval } (\text{Val } 1) (\text{EVAL } (\text{Catch } (\text{Add } (\text{Val } 2) \text{Throw}) (\text{Val } 3))) \text{STOP} \\
= & \text{exec } (\text{EVAL } (\text{Catch } (\text{Add } (\text{Val } 2) \text{Throw}) (\text{Val } 3))) \text{STOP} 1 \\
= & \text{eval } (\text{Catch } (\text{Add } (\text{Val } 2) \text{Throw}) (\text{Val } 3)) (\text{ADD } 1 \text{STOP}) \\
= & \text{eval } (\text{Add } (\text{Val } 2) \text{Throw}) (\text{HAND } (\text{Val } 3) (\text{ADD } 1 \text{STOP})) \\
= & \text{eval } (\text{Val } 2) (\text{EVAL } \text{Throw } (\text{HAND } (\text{Val } 3) (\text{ADD } 1 \text{STOP}))) \\
= & \text{exec } (\text{EVAL } \text{Throw } (\text{HAND } (\text{Val } 3) (\text{ADD } 1 \text{STOP}))) 2 \\
= & \text{eval } \text{Throw } (\text{ADD } 2 (\text{HAND } (\text{Val } 3) (\text{ADD } 1 \text{STOP}))) \\
= & \text{unwind } (\text{ADD } 2 (\text{HAND } (\text{Val } 3) (\text{ADD } 1 \text{STOP}))) \\
= & \text{unwind } (\text{HAND } (\text{Val } 3) (\text{ADD } 1 \text{STOP})) \\
= & \text{eval } (\text{Val } 3) (\text{ADD } 1 \text{STOP}) \\
= & \text{exec } (\text{ADD } 1 \text{STOP}) 3 \\
= & \text{exec } \text{STOP} 4 \\
= & 4
\end{aligned}$$

That is, the machine first proceeds normally by transferring control back and forward between the functions *eval* and *exec*, until the exception is encountered, at which point the control stack is unwound to find the handler expression, and the machine then proceeds normally once again.

## 1.5 FURTHER WORK

We have shown how an abstract machine for a small language with exceptions can be calculated in a systematic way from a semantics for the language, using a three-step process of adding continuations, defunctionalizing, and refactoring. Moreover, the calculations themselves are straightforward, only requiring the basic concepts of structural induction and case analysis.

Possible directions for further work include exploring the impact of higher-level algebraic methods (such as monads [Wad92] and folds [Hut99]) on the calculations, mechanically checking the calculations using a theorem proving system (for example, see [Nip04]), factorising the abstract machine into the composition of a compiler and a virtual machine [ABDM03a], and generalising the underlying language (we are particularly interested in the addition of interrupts.)

## Acknowledgements

Thanks to the referees, Thorsten Altenkirch, Olivier Danvy, Conor McBride, and the WG2.1 meeting in Nottingham for useful comments.

## REFERENCES

- [ABDM03a] Mads Sig Ager, Dariusz Biernacki, Olivier Danvy, and Jan Midtgaard. From Interpreter to Compiler and Virtual Machine: a Functional Derivation. Technical Report RS-03-14, BRICS, Aarhus, Denmark, March 2003.
- [ABDM03b] Mads Sig Ager, Dariusz Biernacki, Olivier Danvy, and Jan Midtgaard. A functional correspondence between evaluators and abstract machines. In *Proceedings of the Fifth ACM-SIGPLAN International Conference on Principles and Practice of Declarative Programming*, Uppsala, Sweden, 2003.
- [ADM04] Mads Sig Ager, Olivier Danvy, and Jan Midtgaard. A functional correspondence between monadic evaluators and abstract machines for languages with computational effects. Research Series RS-04-28, BRICS, Department of Computer Science, University of Aarhus, December 2004.
- [Bac03] Roland Backhouse. *Program Construction: Calculating Implementations from Specifications*. John Wiley, 2003.
- [DN01] Olivier Danvy and Lasse R. Nielsen. Defunctionalization at work. In *Proceedings of the Third ACM-SIGPLAN International Conference on Principles and Practice of Declarative Programming*, Firenze, September 2001.
- [Hue97] Gerard Huet. The Zipper. *Journal of Functional Programming*, 7(5):549–554, September 1997.
- [Hut99] Graham Hutton. A Tutorial on the Universality and Expressiveness of Fold. *Journal of Functional Programming*, 9(4):355–372, July 1999.
- [HW04] Graham Hutton and Joel Wright. Compiling Exceptions Correctly. In *Proceedings of the 7th International Conference on Mathematics of Program Construction*, volume 3125 of *Lecture Notes in Computer Science*, Stirling, Scotland, July 2004. Springer.
- [Lan64] Peter Landin. The mechanical evaluation of expressions. *The Computer Journal*, 6(4):308–320, 1964.
- [Mei92] Erik Meijer. *Calculating Compilers*. PhD thesis, Nijmegen, 1992.
- [Nip04] Tobias Nipkow. Compiling Exceptions Correctly. In *Archive of Formal Proofs*. 2004. Available from <http://afp.sourceforge.net/>.
- [Pey03] Simon Peyton Jones. *Haskell 98 Language and Libraries: The Revised Report*. Cambridge University Press, 2003.
- [Rey72] John C. Reynolds. Definitional Interpreters for Higher-Order Programming Languages. In *Proceedings of the ACM annual conference*, pages 717–740. ACM Press, 1972.
- [Spi90] Mike Spivey. A Functional Theory of Exceptions. *Science of Computer Programming*, 14(1):25–43, 1990.
- [Wad92] Philip Wadler. The Essence of Functional Programming. In *Proc. Principles of Programming Languages*, 1992.